SAS Regression Using Dummy Variables and Oneway ANOVA
/*****************************************************************

 This example illustrates:

      How to create side-by-side boxplots
      How to create dummy variables

      How to use dummy variables in a linear regression model

      How to fit a oneway ANOVA model
 Procs used:

      Proc Means

      Proc Sgplot
      Proc Reg

      Proc Univariate

      Proc GLM
 Filename: regression_lecture2.sas

*******************************************************************/
We first submit a libname statement, pointing to the folder where the SAS dataset, cars.sas7bdat is stored.
OPTIONS FORMCHAR="|----|+|---+=|-/\<>*";

libname b510 "C:\Documents and Settings\kwelch\Desktop\b510";
Next we create a user-defined format to assign descriptions in words to numeric values for ORIGIN.
proc format;

  value originfmt 1="USA"

                  2="Europe"



      3="Japan";

run;
We get descriptive statistics for all variables within each level of ORIGIN by using a class statement, and we apply the user-defined format to display the values of ORIGIN.

proc means data=b510.cars;

  class origin;
  format origin originfmt.;

run;
                                      The MEANS Procedure

              N

   ORIGIN   Obs   Variable     N           Mean        Std Dev        Minimum        Maximum

   -----------------------------------------------------------------------------------------

   USA      253   MPG        248     20.1282258      6.3768059     10.0000000     39.0000000

                  ENGINE     253    247.7134387     98.7799678     85.0000000    455.0000000

                  HORSE      249    119.6064257     39.7991647     52.0000000    230.0000000

                  WEIGHT     253        3367.33    788.6117392        1800.00        5140.00

                  ACCEL      253     14.9284585      2.8011159      8.0000000     22.2000000

                  YEAR       253     75.5217391      3.7145843     70.0000000     82.0000000

                  CYLINDER   253      6.2766798      1.6626528      4.0000000      8.0000000

   Europe    73   MPG         70     27.8914286      6.7239296     16.2000000     44.3000000

                  ENGINE      73    109.4657534     22.3719083     68.0000000    183.0000000

                  HORSE       71     81.0000000     20.8134572     46.0000000    133.0000000

                  WEIGHT      73        2431.49    490.8836172        1825.00        3820.00

                  ACCEL       73     16.8219178      3.0109175     12.2000000     24.8000000

                  YEAR        73     75.7397260      3.5630332     70.0000000     82.0000000

                  CYLINDER    73      4.1506849      0.4907826      4.0000000      6.0000000

   Japan     79   MPG         79     30.4506329      6.0900481     18.0000000     46.6000000

                  ENGINE      79    102.7088608     23.1401260     70.0000000    168.0000000

                  HORSE       79     79.8354430     17.8191991     52.0000000    132.0000000

                  WEIGHT      79        2221.23    320.4972479        1613.00        2930.00

                  ACCEL       79     16.1721519      1.9549370     11.4000000     21.0000000

                  YEAR        79     77.4430380      3.6505947     70.0000000     82.0000000

                  CYLINDER    79      4.1012658      0.5904135      3.0000000      6.0000000

   -----------------------------------------------------------------------------------------

We can see that the mean of vehicle miles per gallon (MPG) is lowest for the American cars, intermediate for the European cars, and highest for the Japanese cars. 

We now look at a side-by-side boxplot of miles per gallon (MPG) for each level of origin.

/*Get side-by-side boxplots of Weight for each

   vehicle origin*/

proc sgplot data=b510.cars;

 vbox mpg/ category=origin;
 format origin originfmt.;
run;

The boxplot shows the pattern of means that we noted in the descriptive statistics. The variance is similar for the American, European and Japanese cars. The distribution of MPG is somewhat positively skewed for American and European cars, and negatively skewed for Japanese cars. There are some high outliers in the American and European cars. Because ORIGIN is a nominal variable, we will not be tempted to think of this as an ordinal relationship. If we had a different coding for ORIGIN, this graph would have shown a different pattern.

Regression Model with Dummy Variables:
Linear regression models were originally developed to link a continuous outcome variable (Y) to continuous predictor variables (X). However, we can extend the model to include categorical predictors by creating a series of dummy variables.
Before we can fit a linear regression model with a categorical predictor (in this case, ORIGIN is nominal) we need to create the dummy variables in a Data Step. We will create three dummy variables, even though only two of them will be used in the regression model. Each dummy variable will be coded as 0 or 1.  A value of 1 will indicate that a case is in a given level of ORIGIN, and a value of 0 will indicate that the case is not in that level of ORIGIN. This is known as "reference level" coding. There are other ways of coding dummy variables, but we will not be using them in this class. The dummy variables for ORIGIN are created in the data step below.
NB: The output shows that the one car with a missing value for ORIGIN does not have a value for any of the dummy variables, as required. Also note that the frequency tabulations show that there is one missing value for each dummy variable. It is necessary to check the coding of dummy variables very carefully to be sure that the number of cases in each category are correctly specified and that missing values are handled correctly!
/*Data step to create dummy variables for each level of ORIGIN*/
data b510.cars2;

  set b510.cars;
  if origin not=. then do;

       American=(origin=1);


 European=(origin=2);


 Japanese=(origin=3);

  end;

run;
proc print data=b510.cars2

  var origin American European Japanese weight;

run;
                   Obs    ORIGIN    American    European    Japanese    WEIGHT

                    1         .        .           .           .         732

                    2    USA           1           0           0        1800

                    3    USA           1           0           0        1875

                    4    USA           1           0           0        1915

                    5    USA           1           0           0        1955
                     . . .

                  256    Europe        0           1           0        1825

                  257    Europe        0           1           0        1834

                  258    Europe        0           1           0        1835

                  259    Europe        0           1           0        1835

                  260    Europe        0           1           0        1845

                     . . .

                  329    Japan         0           0           1        1649

                  330    Japan         0           0           1        1755

                  331    Japan         0           0           1        1760

                  332    Japan         0           0           1        1773

proc freq data=b510.cars2;

  tables origin american european japanese;
  format origin originfmt.;
run;
                                       The FREQ Procedure
                                                     Cumulative    Cumulative

                  ORIGIN    Frequency     Percent     Frequency      Percent

                  -----------------------------------------------------------

                  USA            253       62.47           253        62.47

                  Europe          73       18.02           326        80.49

                  Japan           79       19.51           405       100.00
                                     Frequency Missing = 1

                                                      Cumulative    Cumulative

                 American    Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                        0         152       37.53           152        37.53

                        1         253       62.47           405       100.00
                                     Frequency Missing = 1

                                                      Cumulative    Cumulative

                 European    Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                        0         332       81.98           332        81.98

                        1          73       18.02           405       100.00
                                     Frequency Missing = 1
                                                      Cumulative    Cumulative

                 Japanese    Frequency     Percent     Frequency      Percent

                 -------------------------------------------------------------

                        0         326       80.49           326        80.49

                        1          79       19.51           405       100.00
                                     Frequency Missing = 1

We can now fit a regression model, to predict MPG for each ORIGIN, using dummy variables. We will use American cars as the reference category in this model. To do this we omit the dummy variable for American cars from our model and include the dummy variables for European and Japanese cars. These two dummy variables represent a contrast between the average MPG for European and Japanese cars vs. American cars, respectively. In general, if you have k categories in your categorical variable, you will need to include k-1 dummy variables in the regression model, and omit the dummy variable for the reference category. The model that we will fit is shown below:

Yi = (0 + (1European + (2Japanese + (i
/*Fit a regression model with American cars as the reference category*/
proc reg data=b510.cars2;

  model mpg = european  japanese;

  plot residual.*predicted.;

  output out=regdat p=predicted r=residual rstudent=rstudent;

run; quit;
                                       The REG Procedure

                                         Model: MODEL1

                                    Dependent Variable: MPG

                     Number of Observations Read                        406

                     Number of Observations Used                        397

                     Number of Observations with Missing Values           9

                                      Analysis of Variance

                                             Sum of           Mean

         Source                   DF        Squares         Square    F Value    Pr > F

         Model                     2     7984.95725     3992.47862      97.97    <.0001

         Error                   394          16056       40.75232

         Corrected Total         396          24041

                      Root MSE              6.38375    R-Square     0.3321

                      Dependent Mean       23.55113    Adj R-Sq     0.3287

                      Coeff Var            27.10593

                                      Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1       20.12823        0.40537      49.65      <.0001

              European      1        7.76320        0.86400       8.99      <.0001

              Japanese      1       10.32241        0.82473      12.52      <.0001
We first note that there are 397 observations included in this regression fit. Nine observations were excluded because of missing values in either the outcome variable, MPG, or the predictors (EUROPEAN, JAPANESE). 

Next, we look at the Analysis of Variance table. Always check this table to be sure the model is set up correctly. The Corrected Total df = n-1, which is 396 for this model. The Model df = 2, because we have two dummy variables as predictors. The Error df = 394, which is calculated as: 
Error df = Corrected Total df – Model df.
The F test is reported as F (2, 394)  = 97.97, p <0.0001, and indicates that we have a significant overall model. However, we don't know where the significant differences lie. We will check the parameter estimates table for this.

The Model R-square is 0.3321, indicating that about 33% of the total variance of MPG is explained by this regression model.
The parameter estimate for the Intercept represents the estimated MPG for the reference category, American cars. Compare this estimate (20.128) with the mean MPG for American cars in the descriptive statistics. The parameter estimate for European (7.76) represents the contrast in mean MPG for European cars vs. American cars (the reference). That is, European cars are estimated to have a mean value of MPG that is 7.76 units higher than American cars on average. This difference is significant, t(394) = 8.99, p<0.0001. The parameter estimate for Japanese (10.32) represents the contrast in mean MPG for Japanese cars vs. American cars.  Japanese cars are estimated to have a mean value of MPG that is 10.32 units higher than American cars. This difference is significant, t(394) = 12.52, p<0.0001).

NB: The df (degrees of freedom) shown in the table of Parameter Estimates table are all equal to 1. This means that we are looking at one parameter for each of these estimates, and it is not the df to use for the t-tests. The df to use for the t-tests is the Error df, which, in this case is 394.
We can use the model output to calculate the mean MPG for European cars by adding the estimated intercept plus the parameter estimate for European (20.12823 + 7.76320 = 27.89143). We calculate the mean MPG for Japanese cars by adding the intercept plus the parameter estimate for Japanese (20.12823 + 10.32241 = 30.45064). We can compare these values with the values in the output from Proc Means, and see that they agree within rounding error.

We now look at the residual vs. predicted values plot to see if there is approximate equality of variances across levels of origin. Notice that there is only one predicted value of MPG for each ORIGIN. We can see that the spread of residuals for each origin is approximately the same, indicating that we have reasonable homoskedasticity for this model fit.

We now look at the distribution of the studentized-deleted residuals for this model, using Proc Univariate, to see if we have reasonably normally distributed residuals. The plot indicates that we have somewhat longer tails than expected for a normal distribution, but it is reasonably symmetric.
/*Check distribution of studentized-deleted residuals*/ 
proc univariate data=regdat normal;

  var rstudent;

  histogram;

  qqplot / normal (mu=est sigma=est);

run;

The tests for normality are significant, as shown below. Note that we are testing:

H0: the distribution of residuals is normal

HA: the distribution of residuals is not normal

We reject H0 and conclude that the residuals are not normally distributed. We might wish to investigate transformations of Y (e.g., the log of Y) to get more normally distributed residuals. However, these departures from normality do not appear to be severe, and transformations will not be explored here.

                                      Tests for Normality

                   Test                  --Statistic---    -----p Value------

                   Shapiro-Wilk          W     0.962915    Pr < W     <0.0001

                   Kolmogorov-Smirnov    D     0.083057    Pr > D     <0.0100

                   Cramer-von Mises      W-Sq  0.587669    Pr > W-Sq  <0.0050

                   Anderson-Darling      A-Sq  3.889483    Pr > A-Sq  <0.0050
We now examine the output data set (regdat) produced by Proc Reg (the name we choose for this data set is arbitrary). This new data set will contain all the original variables, plus the new ones that we requested. Again, notice that the predicted value is the same for all observations within the same origin, and is equal to the mean MPG for that level of origin. Also, notice that some of the residuals are positive and some are negative.
/*Take a look at the output data set*/

proc print data=regdat;

  var mpg origin predicted residual rstudent;

run;

                   Obs     MPG    ORIGIN    predicted    residual    rstudent

                     1       9         .       .            .          .

                     2      36    USA        20.1282      15.9718     2.52403

                     3      39    USA        20.1282      18.8718     2.99194

                     4      36    USA        20.1282      15.5718     2.45983

                     5      26    USA        20.1282       5.8718     0.92148

                     6      29    USA        20.1282       8.8718     1.39422

                     7      34    USA        20.1282      14.2718     2.25170

                     8      25    USA        20.1282       4.8718     0.76429

                     9      31    USA        20.1282      10.3718     1.63143

                    10      34    USA        20.1282      13.3718     2.10805
                    . . .

                   101      23     USA       20.1282      2.37177     0.37188

                   102      14     USA       20.1282     -6.12823    -0.96182

                   103      20     USA       20.1282     -0.12823    -0.02010

                   104      18     USA       20.1282     -2.12823    -0.33368

                    . . .

                   308      22    Europe     27.8914      -6.2914    -0.99263

                   309       .    Europe     27.8914        .          .

                   310      20    Europe     27.8914      -7.5914    -1.19843

                   311      19    Europe     27.8914      -8.8914    -1.40461

                   312      18    Europe     27.8914      -9.8914    -1.56351

                   313      22    Europe     27.8914      -5.8914    -0.92938

                   314      36    Europe     27.8914       8.5086     1.34384

                   315      23    Europe     27.8914      -4.8914    -0.77137

                   316      21    Europe     27.8914      -6.8914    -1.08757

                   317       .    Europe     27.8914        .          .

                   318      17    Europe     27.8914     -10.8914    -1.72272

                   319      20    Europe     27.8914      -7.8914    -1.24597

                   320      31    Europe     27.8914       2.8086     0.44268

                   321      27    Europe     27.8914      -0.6914    -0.10896

                   322      28    Europe     27.8914       0.2086     0.03287

                   323      30    Europe     27.8914       2.1086     0.33231

                    . . .

                   362      18    Japan      30.4506     -12.4506    -1.96999

                   363      27    Japan      30.4506      -3.4506    -0.54350

                   364      27    Japan      30.4506      -3.4506    -0.54350

                   365      30    Japan      30.4506      -0.9506    -0.14968

                   366      34    Japan      30.4506       3.3494     0.52754

                   367      28    Japan      30.4506      -2.4506    -0.38592

                   368      36    Japan      30.4506       5.5494     0.87459

                   369      29    Japan      30.4506      -1.4506    -0.22841

                   370      36    Japan      30.4506       5.5494     0.87459

                   371      34    Japan      30.4506       3.2494     0.51178
We now fit a new model, using Japan as the reference category of ORIGIN. This time we include the two dummy variables, AMERICAN and EUROPEAN in our model. The SAS commands and output are shown below:
/*Refit the model using Japan as the reference category*/

proc reg data=b510.cars2;

  model mpg = american european;

  plot residual.*predicted.;

run; quit;
                                       The REG Procedure

                                         Model: MODEL1

                                    Dependent Variable: MPG

                     Number of Observations Read                        406

                     Number of Observations Used                        397

                     Number of Observations with Missing Values           9

                                      Analysis of Variance

                                             Sum of           Mean

         Source                   DF        Squares         Square    F Value    Pr > F

         Model                     2     7984.95725     3992.47862      97.97    <.0001

         Error                   394          16056       40.75232

         Corrected Total         396          24041

                      Root MSE              6.38375    R-Square     0.3321

                      Dependent Mean       23.55113    Adj R-Sq     0.3287

                      Coeff Var            27.10593

                                      Parameter Estimates

                                   Parameter       Standard

              Variable     DF       Estimate          Error    t Value    Pr > |t|

              Intercept     1       30.45063        0.71823      42.40      <.0001

              American      1      -10.32241        0.82473     -12.52      <.0001

              European      1       -2.55920        1.04787      -2.44      0.0150

Notice that the Analysis of Variance table for this model is the same as for the previous model, and the Model R-square is the same. However, the parameter estimates differ, because they represent different quantities than they did in the first model. This is because we have fit the same model, but used a different way to parameterize the dummy variables for ORIGIN. 

The intercept is now the estimated mean MPG for Japanese cars (30.45). The parameter estimate for AMERICAN represents the contrast in the mean MPG for American cars vs. Japanese cars (American cars have on average, 10.32 lower MPG than do Japanese cars). The parameter estimate for EUROPEAN represents the contrast in the mean MPG for European cars vs. Japanese cars (European cars have on average, 2.56 MPG lower MPG than do Japanese cars). 

Oneway ANOVA Model  using Proc GLM:

We can use a oneway ANOVA (analysis of variance) to fit the same linear model as we previously fit using dummy variables in a linear regression. Both ANOVA and linear regression models are examples of GLM, or General Linear Models. The linear regression model with dummy variables and the ANOVA model are equivalent, but have different historical origins. ANOVA models were originally developed in an experimental setting, and are useful ways to compare means of continuous variables (Y) for different levels of categorical predictors (Xs). The focus in the ANOVA model setting is on comparing means of a continuous variable across levels of predictor variables, and the parameter estimates are by default not displayed, unlike linear regression models in which the parameter estimates are the focus of the analysis. You can use an ANOVA model to compare the means of a continuous variable (Y) for a categorical predictor with two or more levels. If you have only two levels for your categorical predictor, the oneway ANOVA model will give equivalent results to an independent samples t-test (assuming equal variances).
The model we investigate here is called a oneway ANOVA because there is only one categorical predictor. You may also fit a twoway or higher-way ANOVA, if you have two or more categorical predictors in the model.
When we use Proc GLM, we do not have to create the dummy variables as we did for Proc Reg. Here is sample SAS code for fitting a oneway ANOVA model using Proc GLM. Note the class statement specifying ORIGIN as a class variable. This causes SAS to create dummy variables for ORIGIN automatically. SAS will use the highest formatted level (USA in this case) of ORIGIN as the reference category. SAS also over-parameterizes the model, including a dummy variable for each level of ORIGIN, but setting the parameter for the highest level equal to zero. 

/*Fit an ANOVA model (USA will be the default reference category)*/

proc glm data=b510.cars2;

  class origin;

  model mpg = origin / solution;

  means origin / hovtest=levene(type=abs) tukey;
  format origin originfmt.;
run; quit;
                                       The GLM Procedure

                                    Class Level Information

                           Class         Levels    Values

                           ORIGIN             3    Europe Japan USA

                            Number of Observations Read         406

                            Number of Observations Used         397

Dependent Variable: MPG
                                              Sum of

      Source                      DF         Squares     Mean Square    F Value    Pr > F
      Model                        2      7984.95725      3992.47862      97.97    <.0001

      Error                      394     16056.41474        40.75232
      Corrected Total            396     24041.37199

                       R-Square     Coeff Var      Root MSE      MPG Mean
                       0.332134      27.10593      6.383755      23.55113

      Source                      DF       Type I SS     Mean Square    F Value    Pr > F
      ORIGIN                       2     7984.957245     3992.478623      97.97    <.0001

      Source                      DF     Type III SS     Mean Square    F Value    Pr > F
      ORIGIN                       2     7984.957245     3992.478623      97.97    <.0001

                                                      Standard

           Parameter                Estimate             Error    t Value    Pr > |t|
           Intercept             20.12822581 B      0.40536882      49.65      <.0001

           ORIGIN    Europe       7.76320276 B      0.86400226       8.99      <.0001

           ORIGIN    Japan       10.32240710 B      0.82472786      12.52      <.0001

           ORIGIN    USA          0.00000000 B       .                .         .

NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to solve

      the normal equations.  Terms whose estimates are followed by the letter 'B' are not

      uniquely estimable.

Note that the ANOVA Table results and Model R-Square here are the same as for the Regression Model with dummy variables.  The parameter estimates (not displayed by default) are the same as those obtained in the dummy variable regression model, with American as the reference category. Here, USA is the reference category, because its format has the highest level of ORIGIN alphabetically.
We also requested Levene's test for homogeneity of variances for the three groups of MPG. Because we are testing H0: (2American  = (2European = (2Japanese, and we do not reject H0, we conclude that the variances are not significantly different from each other. The results of Levene's test indicates that there is not a problem with inequality of variances for this model. 

                          Levene's Test for Homogeneity of MPG Variance

                         ANOVA of Absolute Deviations from Group Means

                                       Sum of        Mean

                 Source        DF     Squares      Square    F Value    Pr > F

                 ORIGIN         2      3.0446      1.5223       0.11    0.8997

                 Error        394      5674.0     14.4009

If we had rejected H0 for this model, we could have instructed SAS to fit a model allowing unequal variances across levels of origin by using the following syntax as part of Proc GLM.

  means origin / hovtest=levene(type=abs) tukey welch;

The Welch test (not shown here) gives a p-value for the ANOVA model, adjusted for unequal variances. This method for a oneway ANOVA is similar to the t-test results for unequal variances using Proc ttest.
The output below is for Tukey's studentized range test for comparing the means of MPG for each pair of origins. There are 3 possible comparisons of means, and the Tukey procedure assures that the overall experimentwise Type I error rate will not be exceeded. By default, SAS uses an overall alpha level of 0.05. 

                          Tukey's Studentized Range (HSD) Test for MPG
                 NOTE: This test controls the Type I experimentwise error rate.

                          Alpha                                   0.05

                          Error Degrees of Freedom                 394

                          Error Mean Square                   40.75232

                          Critical Value of Studentized Range  3.32709

                Comparisons significant at the 0.05 level are indicated by ***.

                                      Difference

                       ORIGIN            Between      Simultaneous 95%

                     Comparison            Means     Confidence Limits
                   Japan  - Europe        2.5592       0.0940    5.0244  ***

                   Japan  - USA          10.3224       8.3821   12.2627  ***

                   Europe - Japan        -2.5592      -5.0244   -0.0940  ***

                   Europe - USA           7.7632       5.7305    9.7959  ***

                   USA    - Japan       -10.3224     -12.2627   -8.3821  ***

                   USA    - Europe       -7.7632      -9.7959   -5.7305  ***
We can see from the above output that all pairwise comparisons of means are significant at the .05 level, after applying the Tukey adjustment for multiple comparisons. SAS shows each comparison of means twice, doing the subtraction of the two means being compared both ways. There are many methods for multiple comparisons available in SAS Proc GLM and other ANOVA procedures, such as Proc Mixed, which we will not cover here. 

